Includes barcode for easy software entry.

HEWLETT-PACKARD

HP-41

USERS' LIBRARY SOLUTIONS

Civil Engineering



NOTICE

The program material contained herein is supplied without representation or warranty of any kind. Hewlett-Packard Company therefore assumes no responsibility and shall have no liability, consequential or otherwise, of any kind arising from the use of this program material or any part thereof.

The product related to this manual is no longer in production at Hewlett Packard Co. As a service to our customers, we are supplying you with a photocopy of the orignal document.

INTRODUCTION

This HP-41C Solutions book was written to help you get the most from your calculator. The programs were chosen to provide useful calculations for many of the common problems encountered.

They will provide you with immediate capabilities in your everyday calculations and you will find them useful as guides to programming techniques for writing your own customized software. The comments on each program listing describe the approach used to reach the solution and help you follow the programmer's logic as you become and expert on your HP calculator.

KEYING A PROGRAM INTO THE HP-41C

There are several things that you should keep in mind while you are keying in programs from the program listings provided in this book. The output from the HP 82143A printer provides a convenient way of listing and an easily understood method of keying in programs without showing every keystroke. This type of output is what appears in this handbook. Once you understand the procedure for keying programs in from the printed listings, you will find this method simple and fast. Here is the procedure:

1. At the end of each program listing is a listing of status information required to properly execute that program. Included is the SIZE allocation required. Before you begin keying in the program, press [XEQ] [ALPHA] SIZE [ALPHA] and specify the allocation (three digits; e.g., 10 should be specified as 010).

   Also included in the status information is the display format and status of flags important to the program. To ensure proper execution, check to see that the display status of the HP-41C is set as specified and check to see that all applicable flags are set or clear as specified.

2. Set the HP-41C to PRGM mode (press the [PRGM] key) and press [■] [GTO] [.] [.] to prepare the calculator for the new program.

3. Begin keying in the program. Following is a list of hints that will help you when you key in your programs from the program listings in this handbook.

   a. When you see " (quote marks) around a character or group of characters in the program listing, those characters are ALPHA. To key them in, simply press [ALPHA], key in the characters, then press [ALPHA] again. So "SAMPLE" would be keyed in as [ALPHA] "SAMPLE" [ALPHA].

   b. The diamond in front of each LBL instruction is only a visual aid to help you locate labels in the program listings. When you key in a program, ignore the diamond.

   c. The printer indication of divide sign is /. When you see / in the program listing, press [÷].

   d. The printer indication of the multiply sign is ⋇. When you see ⋇ in the program listing, press [×].

   e. The ⊢ character in the program listing is an indication of the [APPEND] function. When you see ⊢, press [■] [APPEND] in ALPHA mode (press [■] and the K key).

   f. All operations requiring register addresses accept those addresses in these forms:

      nn (a two-digit number)
      IND nn (INDIRECT: [■], followed fy a two-digit number)
      X, Y, Z, T, or L (a STACK address: [.] followed by X, Y, Z, T, or L)
      IND X, Y, Z, T or L (INDIRECT stack: [■] [.] followed by X, Y, Z, T, or L)

      Indirect addresses are specified by pressing [■] and then the indirect address. Stack addresses are specified by pressing [.] followed by X, Y, Z, T, or L. Indirect stack addresses are specified by pressing [■] [.] and X, Y, Z, T, or L.

Printer Listing	Keystrokes	Display
01◆LBL "SAMPLE"	[■] [LBL] [ALPHA] SAMPLE [ALPHA]	*01 LBL"SAMPLE*
02 "THIS IS A "	[ALPHA] *THIS IS A* [ALPHA]	*02"THIS IS A*
03 "⊢SAMPLE"	[ALPHA] [■] [APPEND] SAMPLE	*03" ⊢ SAMPLE*
04 AVIEW	[■] AVIEW [ALPHA]	*04 AVIEW*
05 6	6	*05 6*
06 ENTER↑	[ENTER↑]	*06 ENTER /*
07 -2	2 [CHS]	*07 -2*
08 /	[÷]	*08 /*
09 ABS	[XEQ] [ALPHA] ABS [ALPHA]	*09 ABS*
10 STO IND L	[STO] [■] [.] L	*10 STO IND L*
11 "R3="	[ALPHA] R3= [■] [ARCL] 03	*11"R3=*
12 ARCL 03	[■] [AVIEW]	*12 ARCL 03*
13 AVIEW	[ALPHA]	*13 AVIEW*
14 RTN	[■] [RTN]	*14 RTN*

TABLE OF CONTENTS

1. STEEL COLUMN FORMULA 1
   Calculates the allowable load and the maximum load for structural steel columns.

2. REINFORCED CONCRETE BEAMS 8
   Design and analysis of reinforced concrete beams.

3. STRESS IN THICK-WALLED CYLINDERS 19
   Calculates the radial and tangential components of normal stress.

4. PROPERTIES OF SPECIAL SECTIONS 25
   Find Ix, Iy, and Area for 5 different section types.

5. COMPRESSIVE BUCKLING 34
   An interchangeable solution for the four properties of slender compression members or columns.

6. VECTORS 40
   Addition, subtraction, cross product, dot product or find the magnitude, 2 or 3 dimensions.

7. BEAMS FIXED AT BOTH ENDS 50
   Calculate deflection, slope, moment, and shear.

8. SIMPLY SUPPORTED BEAMS 61
   Calculate deflection, slope, moment, and shear.

9. CANTILEVER BEAMS 71
   Calculate deflection, slope, moment, and shear.

10. BOLT TORQUE 82
    Calculate the torque needed to supply a specified load.

STEEL COLUMN FORMULA

This program computes the allowable load and the maximum load for structural steel columns using the American Institute of Steel Construction formula (1961). The column ends must be welded, riveted, or otherwise constrained against deflection and rotation.

Equations:



$$P_{allow} = A\,\sigma_y\,[1 - (\ell/k)^2/2\,C^2]/m \qquad \text{for } \ell/k < C$$

$$P_{allow} = A(1.0273 \times 10^{12}\ N/m^2)/(\ell/k)^2 \qquad \text{for } C < \ell/k \leq 200$$

$$C^2 = 2\,\pi^2 E/\sigma_y$$

$$m = 5/3 \times 3(\ell/k)/8C - [(\ell/k)/2C]^3$$

$$P_{max} = P_{allow}\ m$$

Definitions:

P_{allow} is the allowable load;
P_{max} is the maximum load the column could carry;
A is the area of the section;
ℓ is the length of the column;
k is the minimum radius of gyration of the column cross section;
I is the minimum moment of inertia of the cross section;
σ_y is the yield point of the steel.
E is the modulous of elasticity of steel.

Reference:

Roark, Raymond J.; Young, Warren C.; Formulas for Stress and Strain, McGraw-Hill, 1975.

Remarks:

Columns must be nominally straight, homogeneous, and of uniform cross section.

Example 1:

Two steel channels are lased together to form the cross section below:



Calculate the allowable and maximum loads using the following specifications:

$k = 81.0 \times 10^{-3}$ m $A = 9.46 \times 10^{-3}$ m^2 $\sigma_y = 248 \times 10^6$ N/m^2 $\ell = 7.5$ m

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 010	
[XEQ] [ALPHA] METRIC [ALPHA]	YPS?
248 [EEX] 6 [R/S]	A?
9.46 [EEX] 3 [CHS] [R/S]	L?
7.5 [R/S]	K?
81 [EEX] 3 [CHS] [R/S]	Pa=918.2E3
[R/S]	PMAX=1.736E6

User Instructions

SIZE: 010

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program			
2	Begin execution:			
	for metric units		[XEQ] METRIC	YPS?
	for English units		[XEQ] ENGLISH	YPS?
	and input data	σ_y	[R/S]	A?
		area	[R/S]	L?
		length	[R/S]	K?
		K	[R/S]	
3	Allowable load is displayed			Pa =
4	To find maximum load		[R/S]	PMAX =

Program Listings

```
 01◆LBL "MET
RIC"
 02 "YPS?"
 03 PROMPT
 04 STO 09
 05 207 E9
 06 X<>Y
 07 /
 08 PI
 09 X↑2
 10 *
 11 ENTER↑
 12 +
 13 SQRT
 14 STO 03
 15 10273 E8
 16 STO 07
 17 GTO 00
 18◆LBL "ENG
LISH"
 19 "YPS?"
 20 PROMPT
 21 STO 09
 22 30 E6
 23 X<>Y
 24 /
 25 PI
 26 X↑2
 27 *
 28 ENTER↑
 29 +
 30 SQRT
 31 STO 03
 32 149 E6
 33 STO 07
 34◆LBL 00
 35 ENG 3
 36 "A?"
 37 PROMPT
 38 STO 01
 39 "L?"
 40 PROMPT
 41 STO 04
 42 "K?"
 43 PROMPT
 44 STO 05
 45◆LBL A
 46 CF 01
 47 RCL 04
 48 RCL 05
 49 /
```

metric set-up

English set-up

Data promting

Calculate P_{allow}

```
 50 STO 08
 51 RCL 03
 52 X>Y?
 53 GTO 01
 54 X<>Y
 55 200
 56 X<>Y
 57 X>Y?
 58 GTO 05
 59 SF 01
 60◆LBL 01
 61 3
 62 RCL 08
 63 RCL 03
 64 /
 65 X↑2
 66 STO 08
 67 LASTX
 68 RDN
 69 -
 70 R↑
 71 *
 72 8
 73 /
 74 5
 75 ENTER↑
 76 3
 77 /
 78 +
 79 STO 06
 80 FS? 01
 81 GTO 00
 82 RCL 09
 83 X<>Y
 84 /
 85 1
 86 RCL 08
 87 2
 88 /
 89 -
 90 *
 91 GTO 07
 92◆LBL 00
 93 RCL 07
 94 RCL 04
 95 RCL 05
 96 /
 97 X↑2
 98 /
 99◆LBL 07
100 RCL 01
```

Program Listings

```
101 *
102 "Pa="
103 XEQ 05
104 RCL 06
105 *
106 "PMAX="
107◆LBL 05
108 ARCL X
109 AVIEW
110 STOP
111 RTN
112 .END.
```

P_{max}

Display

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00		50	
	A		
	C		
	ℓ		
05	K	55	
	m		
	constant (ℓ/k)		
	temp storage		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 010 TOT. REG. 36 USER MODE

ENG 3 FIX ____ SCI ____ ON ____ OFF X

DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
01		C ≥ ℓ/k	C < ℓ/k

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

STEEL COLUMN FORMULA

PROGRAM REGISTERS NEEDED: 27

HEWLETT PACKARD
SOLUTION BOOK:
CIVIL ENGINEERING



REINFORCED CONCRETE BEAMS

This program can be used in the design and analysis of rectangular reinforced concrete beams in accordance with the strength design method of the American Concrete Institute Code (ACI 318-71). The program solves interchangeably between the following six variables:

- A_s - The area of nonprestressed tension reinforcement (in^2 or cm^2);
- b - The width of the member (in or cm);
- M - The maximum internal bending moment (in-lb or kg-cm);
- d - The depth to the centroid of the reinforcing steel (in or cm);
- f_c - The compressive strength of the concrete (psi or kg/cm^2);
- f_y - The yield strength of the steel (psi or kg/cm^2).



During calculation of the parameters listed above, the calculator checks to be sure that enough reinforcement has been specified to meet the minimum allowable value:

$$\frac{A_s}{bd} > \frac{200}{f_y}$$

If this condition is not met the display will flash 10.50 which signifies that the design does not meet section 10.5 of the ACI code. Stop the flashing by pressing [R/S]. Press [R↓] to see the current value of A_s. Press [R↓] again to see the minimum allowable value of A_s.

The program also checks for too much steel. Code section 10.32 specifies the maximum steel area as:

$$\frac{A_{smax}}{bd} = (0.6375)\ \beta_1\ \frac{f_c}{f_y}\ \frac{87000}{87000 + f_y}$$

where $$\beta_1 = \begin{cases} 0.85 \text{ for } f_c \leq 4000 \\ 0.85 - (f_c - 4000)/20000 \text{ for } f_c > 4000 \end{cases}$$

If too much steel has been specified, the calculator flashes 10.32. Stop the flashing by pressing [R/S], then press [R↓] to see the current steel area. Press [R↓] again to see the maximum allowable tension steel area.

If the program halts displaying "DATA ERROR", the input values are mathematically impossible to satisfy. This may be due to an entry error or the configuration may be mathematically undefined. If this is the case, increase the beam size and/or decrease the moment.

Optionally, the depth of the compression zone (a) may be calculated and the depth of the neutral axis (NA) may be calculated. The depth of the neutral axis is important since T-beams may be modeled as rectangular beams if the slab or flange equals or exceeds the depth of the neutral axis.

Equations:

$$M = d\ \phi\ A_s\ f_y - (0.59\ \phi\ A_s^2\ f_y^2)/(b\ f_c)$$

ϕ = factor of safety = 0.9

Reference:

ACI Standard Building Code Requirements for Reinforced Concrete (ACI 318-71), American Concrete Institute, May 1976 printing.

Remarks:

This program is intended as an aid to computation and cannot replace an understanding of ACI 318-71.

This program does not check for deflection or shear stress modes of failure. Refer to ACI 318-71 for specifics on deflection and shear stress.

Example 1:

For the specifications below, calculate the amount of reinforcing steel required.

$M = 1.2 \times 10^6$ in-lb $b = 18$ in $d = 26$ in $f_c = 3500$ psi

$f_y = 50000$ psi

Keystrokes:	Display:
[USER]	(set USER mode)
[XEQ] [ALPHA] SIZE [ALPHA] 011	
[XEQ] [ALPHA] ENG [ALPHA]	AS?
1 [R/S]	b?
18 [R/S]	M?
1.2 [EEX] 6 [R/S]	d?
26 [R/S]	FC?
3500 [R/S]	FY?
50000 [R/S]	
[A]	10.50 00

(Flashing display indicates that calculated steel area is too small to meet ACI minimum as specified in ACI 10.5. Press [R/S] to halt the flashing display. Press [R↓] to see the calculated value, then press [R↓] again to see the minimum value, then use the minimum value to recalculate M.)

[R/S] [R↓]	1.045 00
[R↓]	1.872 00
[STO] 01	
1 [STO] 03	
[C]	M=2.116E6

User Instructions

SIZE: 011

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program (USER mode)		[USER]	
2	Begin execution			
	English unit		[XEQ] ENG	AS?
	Metric		[XEQ] MET	AS?
	and input values	A_s	[R/S]	b?
	(input 1 for unknown value)	b	[R/S]	M?
		M	[R/S]	d?
		d	[R/S]	FC?
		f_c	[R/S]	FY?
		f_y	[R/S]	
3	Calculate unknown			
	A_s		[A]	AS=
	b		[B]	b=
	M		[C]	M=
	d		[D]	d=
	f_c		[E]	FC=
	f_y		[F]	FY=
4	To change a value, store the new value and			
	place 1.0 in the register of the unknown.			
	A_s	A_s	[STO] 01	
	b	b	[STO] 02	
	M	M	[STO] 03	
	d	d	[STO] 04	
	f_c	f_c	[STO] 05	
	f_y	f_y	[STO] 06	
		1	[STO] (unknown)	
6	To calculate depth of compressive stress			

User Instructions

				SIZE:
STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
	block		[f] [B]	a=
7	To calculate depth of neutral axis		[f] [C]	NA=

Program Listings

```
01◆LBL "ENG
"
02 CF 00
03 GTO 00
04◆LBL "MET
"
05 SF 00
06◆LBL 00
07 ENG 3
08 "AS?"
09 PROMPT
10 STO 01
11 "b?"
12 PROMPT
13 STO 02
14 "M?"
15 PROMPT
16 STO 03
17 "d?"
18 PROMPT
19 STO 04
20 "FC?"
21 PROMPT
22 STO 05
23 "FY?"
24 PROMPT
25 STO 06
26 STOP
27◆LBL 03
28 .59
29 RCL 05
30 RCL 02
31 *
32 /
33 RCL 01
34 RCL 06
35 *
36 *
37 LASTX
38 .9
39 *
40 *
41 LASTX
42 RCL 04
43 *
44 RTN
45◆LBL A
46 XEQ 03
47 "AS="
48 1
49 STO 10
50 GTO 01
51◆LBL F
52 XEQ 03
53 "FY="
54 6
55 STO 10
56◆LBL 01
57 RDN
58 CHS
59 ENTER↑
60 RDN
61 X<>Y
62 R↑
63 X↑2
64 X<>Y
65 STO 00
66 4
67 *
68 RCL 03
69 *
70 -
71 SQRT
72 +
73 RCL 00
74 /
75 2
76 /
77 CHS
78 GTO 00
79◆LBL C
80 XEQ 03
81 "M="
82 3
83 STO 10
84 RDN
85 X<>Y
86 -
87 GTO 00
88◆LBL B
89 XEQ 03
90 "b="
91 2
92 STO 10
93 GTO 01
94◆LBL E
95 XEQ 03
96 "FC="
97 5
98 STO 10
99◆LBL 01
100 RDN
```

Initialization

Common to all solutions

A_s

f_y

M

b

fc

Program Listings

```
101 RCL 03
102 -
103 /
104 GTO 00
105◆LBL D
106 XEQ 03
107 "d="
108 4
109 STO 10
110 RDN
111 X<>Y
112 RCL 03
113 +
114 X<>Y
115 /
116◆LBL 00
117 STO IND
10
118 281
119 STO 09
120 6117
121 STO 08
122 14.06
123 FS? 00
124 GTO 00
125 4 E3
126 STO 09
127 87 E3
128 STO 08
129 200
130◆LBL 00
131 RCL 06
132 /
133 RCL 01
134 RCL 02
135 RCL 04
136 *
137 /
138 X>Y?
139 GTO 00
140 RDN
141 LASTX
142 *
143 XEQ 08
144 +
145 RCL 01
146 10.5
147◆LBL 07
148 PSE
149 GTO 07
150◆LBL 00
```

d

Store constants

Check for minimum reinforcing

flashing display

check for too much steel

```
151 .85
152 RCL 05
153 RCL 09
154 -
155 2 E4
156 /
157 X<0?
158 CLX
159 -
160 STO 07
161 .6375
162 *
163 RCL 05
164 *
165 RCL 06
166 /
167 RCL 08
168 RCL 06
169 +
170 /
171 RCL 08
172 *
173 X>Y?
174 GTO 00
175 RCL 02
176 RCL 04
177 *
178 *
179 XEQ 08
180 -
181 RCL 01
182 10.32
183 GTO 07
184◆LBL 00
185 RCL IND
10
186 ARCL X
187 AVIEW
188 STOP
189◆LBL b
190 SF 02
191 GTO 00
192◆LBL c
193 CF 02
194◆LBL 00
195 RCL 01
196 1.18
197 *
198 RCL 02
199 /
200 RCL 06
```

Display

Calculate NA or a

Program Listings

```
201 *
202 RCL 05
203 /
204 FS? 02
205 GTO 00
206 RCL 07
207 /
208 "NA="
209 ARCL X
210 AVIEW
211 STOP
212◆LBL 00
213 "a="
214 ARCL X
215 AVIEW
216 STOP
217◆LBL 08
218 SF 03
219 1 E-4
220 %
221 RTN
222 .END.
```

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	temporary storage	50	
	As		
	b		
	M		
	d		
05	FC	55	
	FY		
	1		
	87000 or 6117		
	4000 or 281		
10	control	60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 011 TOT. REG. 62 USER MODE
ENG 3 FIX ___ SCI ___ ON X OFF ___
DEG ___ RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
00		metric	english
02		a	NA

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

REINFORCED CONCRETE BEAMS

PROGRAM REGISTERS NEEDED: 52

HEWLETT PACKARD
SOLUTION BOOK:
CIVIL ENGINEERING

ROW 1 (1 - 4)

ROW 2 (4 - 8)

ROW 3 (8 - 16)

ROW 4 (17 - 23)

ROW 5 (23 - 32)

ROW 6 (33 - 44)

ROW 7 (45 - 50)

ROW 8 (51 - 57)

ROW 9 (58 - 70)

ROW 10 (71 - 80)

ROW 11 (80 - 88)

ROW 12 (89 - 95)

ROW 13 (95 - 103)

ROW 14 (104 - 110)

ROW 15 (111 - 120)

ROW 16 (120 - 124)

ROW 17 (125 - 130)

ROW 18 (131 - 142)


ROW 19 (143 - 149)
ROW 20 (150 - 158)
ROW 21 (159 - 167)
ROW 22 (168 - 179)
ROW 23 (179 - 185)
ROW 24 (185 - 192)
ROW 25 (193 - 201)
ROW 26 (202 - 209)
ROW 27 (209 - 218)
ROW 28 (218 - 222)

STRESS IN THICK-WALLED CYLINDERS

This program calculates the radial and tangential components of normal stress for thick-walled, cylindrical, pressure vessels.



Equations:

$$s_r = \frac{r_i^2 P_i - r_o^2 P_o}{r_o^2 - r_i^2} - \frac{r_i^2 r_o^2 (P_i - P_o)}{r^2 (r_o^2 - r_i^2)}$$

$$s_t = \frac{r_i^2 P_i - r_o^2 P_o}{r_o^2 - r_i^2} + \frac{r_i^2 r_o^2 (P_i - P_o)}{r^2 (r_o^2 - r_i^2)}$$

where:

- s_r is the radial component of stress;
- s_t is the tangential component of stress;
- r_i is the internal radius;
- r_o is the outer radius;
- r is the radius where calculated stresses occur;
- P_i is the internal pressure;
- P_o is the outside pressure.

Note: A negative stress indicates compression.

Reference:

J.E. Shigley
Mechanical Engineering Design, McGraw Hill, 1963.

Example:

A cylinder has an inner radius of 1.00 inch and an outer radius of 2.00 inches. The inner pressure is 10,000 pounds per square inch and outer pressure is 150 pounds per square inch. What are the values of radial and tangential stresses for radii of 1.00, 1.25, 1.75 and 2.00 inches?

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 007	
[XEQ] [ALPHA] CYL [ALPHA]	RI?
1 [R/S]	PI?
10000 [R/S]	RO?
2 [R/S]	PO?
150 [R/S]	R?
1 [R/S]	SR=-10.00E3
[R/S]	ST=16.27E3
[R/S]	R?
1.25 [R/S]	SR=-5.272E3
[R/S]	ST=11.54E3
[R/S]	R?
1.75 [R/S]	SR=-1.155E3
[R/S]	ST=7.422E3
[R/S]	R?
2 [R/S]	SR=-150.0E0
[R/S]	ST=6.417E3

User Instructions

SIZE: 007

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program			
2	Begin execution and input data		[XEQ] CYL	RI?
		r_i	[R/S]	PI?
		P_i	[R/S]	RO?
		r_o	[R/S]	PO?
		P_o	[R/S]	R?
3		r	[R/S]	SR=
			[R/S]	ST=
4	For new r		[R/S]	[illegible]?
	Go to step 3			

Program Listings

```
01◆LBL "CYL
"
02 ENG 3
03 "RI?"
04 PROMPT
05 X↑2
06 STO 00
07 "PI?"
08 PROMPT
09 STO 01
10 *
11 "RO?"
12 PROMPT
13 X↑2
14 STO 02
15 "PO?"
16 PROMPT
17 STO 03
18 *
19 -
20 STO 06
21 RCL 02
22 RCL 00
23 -
24 STO 04
25 RCL 00
26 RCL 02
27 *
28 RCL 01
29 RCL 03
30 -
31 *
32 STO 05
33◆LBL "R"
34 RCL 06
35 RCL 04
36 /
37 RCL 05
38 "R?"
39 PROMPT
40 X↑2
41 /
42 RCL 04
43 /
44 -
45 "SR="
46 XEQ d
47 STOP
48 LASTX
49 2
50 *
51 +
52 "ST="
53◆LBL d
54 ARCL X
55 AVIEW
56 RTN
57 GTO "R"
58 .END.
```

$r_i^2 Pi - r_o^2 Po$

$r_i^2 r_o^2 (Pi - Po)$

SR

ST

00 70 80 90

Display

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	r_i	50	
	P_i		
	r_o		
	P_o		
	$r_o^2 - r_i^2$		
05	$r_i^2 r_o^2 (P_i - P_o)$	55	
	$r_i^2 P_i - r_o^2 P_o$		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 007 TOT. REG. 20 USER MODE

ENG 3 FIX ____ SCI ____ ON ____ OFF X

DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

STRESS IN THICK-WALLED
CYLINDERS
PROGRAM REGISTERS NEEDED: 14

HEWLETT PACKARD
SOLUTION BOOK:
CIVIL ENGINEERING



PROPERTIES OF SPECIAL SECTIONS

For rectangles, triangles, ellipses, circles, and concentric circles, this program performs an interchangeable solution between the section dimensions and the principle moment of inertia about the x axis. The section area and the principle moment of inertia about the y axis may also be calculated.

Sections and Equations:



$I_x = a^3b/12$

$I_y = ab^3/12$

$A = ab$



$I_x = a^3b/36$

$I_y = ab^3/36$

$A = ab/2$

$I_x = \pi a^3 b/64$

$I_y = \pi ab^3/64$

$A = \pi ab/4$



$I_x = \frac{\pi a^4}{64} = I_y$

$A = \pi a^2/4$



$I_x = \frac{\pi(a^4 - b^4)}{64} = I_y$

$A = \frac{\pi(a^2 - b^2)}{4}$

Example 1:

For the rectangular section below, what is the moment of inertia about the x axis? What is the moment of inertia about the y axis?



Keystrokes:	Display:
[USER]	(set USER mode)
[XEQ] [ALPHA] SIZE [ALPHA] 004	
[XEQ] [ALPHA] REC [ALPHA]	a?
90 [R/S]	b?
42.5 [R/S]	IX?
[R/S]	
[C]	IX=2.582E6
[E]	IY=575.7E3

Example 2:

What is the moment of inertia about the x-axis for the circle below? What is the area?



Keystrokes:	Display:
[XEQ] [ALPHA] CIR [ALPHA]	a?
56 [R/S]	b?
[R/S]	
[R/S]	
[C]	IX=482.7E3
[D]	A=2.463E3

User Instructions

SIZE: 004

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program (USER mode)		[USER]	
2	Execute appropriate program:			
	for rectangle		[XEQ] REC	a?
	for triangle		[XEQ] TRI	a?
	for ellipse		[XEQ] ELL	a?
	for circles		[XEQ] CIR	a?
	for concentric circles		[XEQ] CC	a?
3	Input any two*:	a	[R/S]	b?
	(skip unknown with [R/S])	b	[R/S]	IX?
		Ix	[R/S]	
4	Compute unknown			
	a		[A]	a =
	b		[B]	b =
	Ix		[C]	IX =
5	To compute area:		[D]	A =
6	To compute Iy:		[E]	IY =
	*for circles enter either a or Ix only			
	skipping b with [R/S]			

Program Listings

```
 01◆LBL "REC
"
 02 CF 00
 03 1
 04 GTO 00
 05◆LBL "TRI
"
 06 CF 00
 07 2
 08 GTO 00
 09◆LBL "ELL
"
 10 CF 00
 11 3
 12 GTO 00
 13◆LBL "CIR
"
 14 SF 00
 15 4
 16 GTO 00
 17◆LBL "CC"
 18 CF 00
 19 5
 20◆LBL 00
 21 ENG 3
 22 STO 00
 23 "a?"
 24 PROMPT
 25 STO 01
 26 CLX
 27 "b?"
 28 PROMPT
 29 STO 02
 30 CLX
 31 "IX?"
 32 PROMPT
 33 STO 03
 34 STOP
 35◆LBL A
 36 RCL 03
 37 RCL 02
 38 X≠0?
 39 /
 40 GTO IND
00
 41◆LBL 01
 42 12
 43 GTO 00
 44◆LBL 02
 45 36
 46 GTO 00
```

Initialization

Calculate a

```
 47◆LBL 03
 48 PI
 49 /
 50 64
 51◆LBL 00
 52 *
 53 3
 54 1/X
 55 Y↑X
 56 STO 01
 57 "a="
 58 GTO 10
 59◆LBL 04
 60◆LBL 05
 61 RCL 03
 62 64
 63 *
 64 PI
 65 /
 66 RCL 02
 67 X↑2
 68 X↑2
 69 +
 70 SQRT
 71 SQRT
 72 STO 01
 73 "a="
 74 GTO 10
 75◆LBL B
 76 GTO IND
00
 77◆LBL 01
 78◆LBL 02
 79◆LBL 03
 80 RCL 03
 81 12
 82 *
 83 RCL 01
 84 X↑2
 85 LASTX
 86 *
 87 /
 88 XEQ IND
00
 89 STO 02
 90 "b="
 91 GTO 10
 92◆LBL 02
 93 3
 94 *
 95◆LBL 01
```

Calculate b

Program Listings

```
 96 RTN
 97◆LBL 03
 98 3
 99 /
100 16
101 *
102 PI
103 /
104 "b="
105 GTO 10
106◆LBL 05
107 RCL 01
108 X↑2
109 X↑2
110 RCL 03
111 64
112 *
113 PI
114 /
115 -
116 SQRT
117 SQRT
118 STO 02
119 "b="
120 GTO 10
121◆LBL C
122 RCL 02
123 RCL 01
124 XEQ 09
125 STO 03
126 "IX="
127 GTO 10
128◆LBL E
129 RCL 01
130 RCL 02
131 XEQ 09
132 "IY="
133 GTO 10
134◆LBL 09
135 ENTER↑
136 X↑2
137 GTO IND
00
138◆LBL 01
139 *
140 *
141 12
142 /
143 RTN
144◆LBL 02
145 *
```

Calculate I_x

Calculate I_y

I_x or I_y

```
146 *
147 36
148 /
149 RTN
150◆LBL 03
151 *
152 *
153 PI
154 *
155 64
156 /
157 RTN
158◆LBL 04
159◆LBL 05
160 X<>Y
161 X↑2
162 *
163 X<>Y
164 X↑2
165 X↑2
166 -
167 ABS
168 PI
169 *
170 64
171 /
172 RTN
173◆LBL D
174 RCL 01
175 RCL 02
176 *
177 GTO IND
00
178◆LBL 02
179 2
180 /
181◆LBL 01
182 GTO 00
183◆LBL 03
184 PI
185 *
186 4
187 /
188 GTO 00
189◆LBL 04
190◆LBL 05
191 RCL 01
192 X↑2
193 RCL 02
194 X↑2
195 -
196 PI
```

Calculate Area

Program Listings

```
197 *
198 4
199 /
200◆LBL 00
201 "A="
202◆LBL 10
203 ARCL X
204 AVIEW
205 STOP
206 .END.
```

Display

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	section code	50	
	a		
	b		
	Ix		
05		55	
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 004 TOT. REG. 48 USER MODE

ENG 3 FIX ___ SCI ___ ON X OFF ___

DEG ___ RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
		NONE	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

PROPERTIES OF SPECIAL SECTIONS

PROGRAM REGISTERS NEEDED: 45




ROW 19 (144 - 155)
ROW 20 (155 - 167)
ROW 21 (168 - 177)
ROW 22 (178 - 188)
ROW 23 (189 - 201)
ROW 24 (201 - 206)

COMPRESSIVE BUCKLING

This program performs an interchangeable solution for the four properties of slender compression members or columns: P_{cr}, the critical buckling load; E, the modulus of elasticity; I, the minimum moment of inertia; and ℓ, the length of the member.

Equations:

Three configurations are possible, identified by the number of fixed ends on the member: 0, both ends hinged; 1, one end free and one fixed; 2, both ends fixed.



Remarks:

Uncertainties such as the amount of restraint at the ends, eccentricity of the load, initial warp, nonhomogeneity of the material and deflection caused by lateral loads, can cause very significant changes in the behavior of a compressive member.

Example 1:

If an 8 inch steel ($E = 30 \times 10^6$ psi) piston rod (a piston rod has zero fixed ends) must withstand a load of 15000 pounds without buckling, what moment of inertia must it have?

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 007	
[XEQ] [ALPHA] COMPRES [ALPHA]	GEOMETRY?
0 [R/S]	P
15000 [R/S]	E
30 [EEX] 6 [R/S]	I
[R/S]	L
8 [R/S]	I=3.242E-3

User Instructions

SIZE: 007

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program			
2	Begin execution (key [R/S] for unknown)		[XEQ] COMPRES	GEOMETRY?
	Key in geometry	0, 1 or 2	[R/S]	P
	vertical load	P	[R/S]	E
	modulus of elasticity	E	[R/S]	I
	moment of inertia	I	[R/S]	L
	length of column	ℓ	[R/S]	(Unknown)=

Program Listings

```
 01◆LBL "COM
PRES"
 02 ENG 3
 03 SF 01
 04 CF 02
 05 "GEOMETR
Y?"
 06 PROMPT
 07 STO 00
 08 "P"
 09 PROMPT
 10 X=0?
 11 XEQ b
 12 STO 01
 13 CLX
 14 "E"
 15 PROMPT
 16 X=0?
 17 XEQ a
 18 STO 02
 19 CLX
 20 "I"
 21 PROMPT
 22 X=0?
 23 XEQ a
 24 STO 03
 25 CLX
 26 "L"
 27 PROMPT
 28 X=0?
 29 XEQ b
 30 STO 04
 31 1
 32 -
 33 X=0?
 34 SF 02
 35 CF 00
 36 "="
 37 ASTO 05
 38 GTO IND
00
 39◆LBL 01
 40 SF 00
 41 XEQ 00
 42 4
 43 /
 44 GTO d
 45◆LBL 02
 46 SF 00
 47 XEQ 00
 48 4
 49 *
 50 GTO d
 51◆LBL 00
 52 PI
 53 X↑2
 54 RCL 02
 55 *
 56 RCL 03
 57 *
 58 RCL 01
 59 /
 60 RCL 04
 61 X↑2
 62 /
 63 FS? 00
 64 RTN
 65◆LBL d
 66 FS? 01
 67 1/X
 68 FS? 02
 69 SQRT
 70 CLA
 71 ARCL 06
 72 ARCL 05
 73 ARCL X
 74 AVIEW
 75 STOP
 76◆LBL b
 77 CF 01
 78◆LBL a
 79 ASTO 06
 80 1
 81 RTN
 82 .END.
```

Initialization

one fixed end

2 fixed ends

No fixed ends

Display

unknown

90

00

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	output label	50	
	P		
	E		
	I		
	ℓ		
05	"="	55	
	unknown		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 007 TOT. REG. 27 USER MODE

ENG 3 FIX ___ SCI ___ ON ___ OFF X

DEG ___ RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
01		invert	don't invert

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

COMPRESSIVE BUCKLING

PROGRAM REGISTERS NEEDED: 21

HEWLETT PACKARD
SOLUTION BOOK:
CIVIL ENGINEERING

ROW 1 (1 - 2)

ROW 2 (3 - 5)

ROW 3 (5 - 14)

ROW 4 (14 - 23)

ROW 5 (23 - 31)

ROW 6 (32 - 39)

ROW 7 (40 - 46)

ROW 8 (47 - 55)

ROW 9 (56 - 66)

ROW 10 (66 - 74)

ROW 11 (75 - 82)

ROW 12 (82 - 82)

VECTOR OPERATIONS

This program performs the basic vector operations of addition, cross product, and dot or scalar product. It also allows conversion between spherical and cartesian coordinates and can find the angle between two vectors.

Equations:



Coordinate conversions:

$x = r \sin \phi \cos \theta$ $\qquad$ $r = \sqrt{x^2 + y^2 + z^2}$

$y = r \sin \phi \sin \theta$ $\qquad$ $\theta = \tan^{-1} (y/x)$

$z = r \cos \phi$ $\qquad$ $\phi = \cos^{-1} (z/ \sqrt{x^2 + y^2 + z^2})$

Vector addition:

$$\bar{V}_1 + \bar{V}_2 = (x_1 + x_2)\ \bar{i} + (y_1 + y_2)\ \bar{j} + (z_1 + z_2)\ k$$

Cross product:

$$\bar{V}_1 \times \bar{V}_2 = (y_1 z_2 - z_1 y_2)\ \bar{i} + (z_1 x_2 - x_1 z_2)\ \bar{j} + (x_1 y_2 - y_1 x_2)\ \bar{k}$$

Dot or scalar product:

$$\bar{V}_1 \cdot \bar{V}_2 = x_1 x_2 + y_1 y_2 + z_1 z_2$$

Angle between vectors:

$$\lambda = \cos^{-1}\left(\frac{V_1 \cdot V_2}{|V_1| \quad |V_2|}\right)$$

Example:

An aircraft flies a heading of 212 degrees at 225 knots. The wind is reported at 20 knots and 140 degrees (which translates to 20 knots and 320 degrees since winds are reported opposite to the direction they blow). What is the course of the aircraft? What is the ground speed?



Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 008	
[XEQ] [ALPHA] SPH [ALPHA]	R?
225 [R/S]	PHI?
90 [R/S]	THETA?
212 [R/S]	R?
20 [R/S]	PHI?
90 [R/S]	THETA?
320 [R/S]	
[XEQ] [ALPHA] ADD [ALPHA]	R=219.64
[R/S]	PHI=90.00
[R/S]	THETA=216.97

User Instructions

SIZE: 008

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program			
2	Begin execution:			
	for rectangular coordinates		[XEQ] REC	X?
	input x_1,y_1,z_1,x_2,y_2,z_2	x_1	[R/S]	Y?
		y_1	[R/S]	Z?
		z_1	[R/S]	X?
		x_2	[R/S]	Y?
		y_2	[R/S]	Z?
		z_2	[R/S]	
	Go to step 3			
	for spherical coordinates		[XEQ] SPH	R?
	input $R_1,\phi_1,\theta_1,R_2,\phi_2,\theta_2$			PHI?
				THETA?
				R?
				PHI?
				THETA?
	Go to step 3			
3	Execute operation			
	addition (then go to step 4)		[XEQ] ADD	
	subtraction (then go to step 4)		[XEQ] SUB	
	dot product		[XEQ] DOT	d=
	angle between		[R/S]	∠=
	cross product (then go to step 4)		[XEQ] CROSS	
4	Answers are displayed:			X= (or) R=
			[R/S]	Y= (or) PHI=
			[R/S]	Z= (or) THETA=

User Instructions

				SIZE:
STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
5	For magnitude of a vector in rectangular			
	coordinates		[XEQ] REC	X?
	input X, Y, Z	X	[R/S]	Y?
		Y	[R/S]	Z?
		Z	[R/S]	X?
			[XEQ] MAG	MAG =

Program Listings

```
 01◆LBL "SPH
"
 02 SF 00
 03 GTO 00
 04◆LBL "REC
"
 05 CF 00
 06◆LBL 00
 07 CLRG
 08 FIX 2
 09 .1
 10 STO 06
 11 XEQ 01
 12 XEQ 01
 13 FC? 00
 14 STOP
 15 RCL 02
 16 COS
 17 RCL 01
 18 SIN
 19 RCL 00
 20 *
 21 *
 22 LASTX
 23 RCL 02
 24 SIN
 25 *
 26 RCL 00
 27 RCL 01
 28 COS
 29 *
 30 STO 02
 31 RDN
 32 STO 01
 33 RDN
 34 STO 00
 35 RCL 05
 36 COS
 37 RCL 04
 38 SIN
 39 RCL 03
 40 *
 41 *
 42 LASTX
 43 RCL 05
 44 SIN
 45 *
 46 RCL 03
 47 RCL 04
 48 COS
 49 *
```

SF00 for SPHERICAL

CF00 for Rectangular

input initialization

spherical to rectangular conversion

```
 50 STO 05
 51 RDN
 52 STO 04
 53 RDN
 54 STO 03
 55 STOP
 56◆LBL 01
 57 "R?"
 58 FC? 00
 59 "X?"
 60 PROMPT
 61 STO IND
06
 62 ISG 06
 63 "PHI?"
 64 FC? 00
 65 "Y?"
 66 PROMPT
 67 STO IND
06
 68 ISG 06
 69 "THETA?"
 70 FC? 00
 71 "Z?"
 72 PROMPT
 73 STO IND
06
 74 ISG 06
 75 RTN
 76◆LBL "ADD
"
 77 CF 02
 78 GTO 01
 79◆LBL "SUB
"
 80 SF 02
 81◆LBL 01
 82 RCL 00
 83 RCL 03
 84 FS? 02
 85 CHS
 86 +
 87 STO 00
 88 RCL 01
 89 RCL 04
 90 FS? 02
 91 CHS
 92 +
 93 STO 01
 94 RCL 02
 95 RCL 05
```

input prompting

CF02 for addition

SF02 for subtraction

addition or subtraction

Program Listings

```
 96 FS? 02
 97 CHS
 98 +
 99 STO 02
100 GTO 00
101◆LBL "DOT
"
102 RCL 00
103 RCL 03
104 *
105 RCL 01
106 RCL 04
107 *
108 +
109 RCL 02
110 RCL 05
111 *
112 +
113 "d="
114 XEQ 04
115 RCL 00
116 RCL 01
117 RCL 02
118 SF 03
119 XEQ 03
120 /
121 RCL 03
122 RCL 04
123 RCL 05
124 XEQ 03
125 /
126 ACOS
127 "∠="
128 GTO 04
129◆LBL "CRO
SS"
130 RCL 01
131 RCL 05
132 *
133 RCL 04
134 RCL 02
135 *
136 -
137 STO 07
138 RCL 02
139 RCL 03
140 *
141 RCL 00
142 RCL 05
143 *
144 -
```

Dot Product

Angle between two vectors

Cross Product

```
145 RCL 00
146 RCL 04
147 *
148 RCL 01
149 RCL 03
150 *
151 -
152 STO 02
153 RDN
154 STO 01
155 RCL 07
156 STO 00
157 GTO 00
158◆LBL "MAG
"
159 CF 03
160 RCL 00
161 RCL 01
162 RCL 02
163◆LBL 03
164 X↑2
165 X<>Y
166 X↑2
167 +
168 X<>Y
169 X↑2
170 +
171 SQRT
172 FS? 03
173 RTN
174 "M="
175 GTO 04
176◆LBL 00
177 CF 01
178 FC? 00
179 GTO 00
180 SF 24
181 RCL 00
182 RCL 01
183 RCL 02
184 SF 03
185 XEQ 03
186 STO 07
187 X=0?
188 1 E-99
189 RCL 02
190 X<>Y
191 /
192 ACOS
193 RCL 00
194 X=0?
```

Magnitude

Program Listings

```
195 1 E-99
196 RCL 01
197 X<>Y
198 X<0?
199 SF 01
200 /
201 ATAN
202 STO 02
203 RDN
204 STO 01
205 RCL 07
206 STO 00
207 CF 24
208 FC? 01
209 GTO 00
210 180
211 ST+ 02
212◆LBL 00
213 RCL 00
214 "R="
215 FC? 00
216 "X="
217 XEQ 04
218 RCL 01
219 "PHI="
220 FC? 00
221 "Y="
222 XEQ 04
223 RCL 02
224 "THETA="
225 FC? 00
226 "Z="
227◆LBL 04
228 ARCL X
229 AVIEW
230 STOP
231 RTN
232 .END.
```

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	x_1	50	
	y_1		
	z_1		
	x_2		
	y_2		
05	z_2	55	
	temp storage		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 008 TOT. REG. 63 USER MODE

ENG ____ FIX 2 SCI ____ ON ____ OFF X

DEG X RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
00		spherical	rectangular
01		+180	normal
02		subtract	add
03		return	continue

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY
NONE			

VECTOR OPERATIONS

PROGRAM REGISTERS NEEDED: 56

HEWLETT PACKARD
SOLUTION BOOK:
CIVIL ENGINEERING

ROW 1 (1 - 4)

ROW 2 (4 - 9)

ROW 3 (10 - 17)

ROW 4 (18 - 30)

ROW 5 (31 - 43)

ROW 6 (44 - 56)

ROW 7 (57 - 62)

ROW 8 (63 - 67)

ROW 9 (68 - 71)

ROW 10 (71 - 76)

ROW 11 (76 - 80)

ROW 12 (80 - 90)

ROW 13 (91 - 101)

ROW 14 (101 - 108)

ROW 15 (109 - 117)

ROW 16 (118 - 125)

ROW 17 (126 - 129)

ROW 18 (129 - 140)


ROW 19 (141 - 153)
ROW 20 (154 - 159)
ROW 21 (159 - 171)
ROW 22 (172 - 178)
ROW 23 (179 - 186)
ROW 24 (187 - 195)
ROW 25 (195 - 203)
ROW 26 (204 - 211)
ROW 27 (211 - 217)
ROW 28 (217 - 222)
ROW 29 (222 - 226)
ROW 30 (226 - 232)

BEAMS FIXED AT BOTH ENDS

This program calculates deflection, slope, moment and shear at any specified point along a beam of uniform cross section, fixed at both ends. Distributed loads, point loads, applied moments or combinations of all three may be modeled. By using the principle of superposition, complicated beams with multiple point loads, and multiple applied moments can be analyzed.

Equations:



$$y = y_1 + y_2 + y_3 \qquad \text{(total deflection)}$$

$$y_1 = \frac{P(\ell - a)^2 x^2}{6EI\ell^3}\,[x(\ell+2a) - 3a\ell]^* \qquad \text{(deflection due to point load)}$$

$$y_2 = \frac{Wx^2}{24EI}\,[x(2\ell-x) - \ell^2] \qquad \text{(distributed load)}$$

$$y_3 = \frac{M(\ell - c)x^2}{\ell^2 EI}\left[\frac{cx}{\ell} + \frac{\ell - 3c}{2}\right]^{**} \qquad \text{(applied moment)}$$

$$\theta = \theta_1 + \theta_2 + \theta_3 \qquad \text{(total slope)}$$

$$\theta_1 = \frac{P(\ell - a)^2 x}{2EI\ell^3}\,[x(\ell+2a) - 2a\ell]^* \qquad \text{(slope due to point load)}$$

$$\theta_2 = \frac{Wx}{12EI}\,[x(3\ell - 2x) - \ell^2] \qquad \text{(slope due to distributed load)}$$

$$\theta_3 = \frac{M(\ell - c)x}{\ell^2 EI}\left[\frac{3cx}{\ell} + \ell - 3c\right]^{**} \qquad \text{(slope due to applied moment)}$$

$$M_x = M_{x1} + M_{x2} + M_{x3} \qquad \text{(total moment)}$$

$$M_{x1} = \frac{P(\ell - a)^2}{\ell^3}\,[x(\ell + 2a) - a\ell]^{*} \qquad \text{(moment due to point load)}$$

$$M_{x2} = \frac{W}{12}\,[6x(\ell - x) - \ell^2] \qquad \text{(moment due to distributed load)}$$

$$M_{x3} = \frac{M(\ell - c)}{\ell^2}\left[\frac{6cx}{\ell} + \ell - 3c\right]^{**} \qquad \text{(moment due to applied moment)}$$

$$V = V_1 + V_2 + V_3 \qquad \text{(total shear)}$$

$$V_1 = \frac{P(\ell - a)^2}{\ell^3}\,(\ell + 2a) \qquad \text{(shear due to point load)}$$

$$V_2 = \frac{-W}{2}\,(2x - \ell) \qquad \text{(shear due to distributed load)}$$

$$V_3 = \frac{-6M(\ell - c)\,c^{**}}{\ell^3} \qquad \text{(shear due to applied moment)}$$

where:

y is the deflection at a distance x from the left support;

θ is the slope (change in y per change in x) at x;

M_x is the moment at x;

V is the shear at x;

I is the moment of inertia of the beam;

E is the modulus of elasticity of the beam;

ℓ is the length of the beam;

P is a concentrated load;

W is a uniformly distributed load with dimensions of force per unit length;

M is an applied moment;

a is the distance from the left support to the point load;

c is the distance to the applied moment.

*If x is greater than a, a is replaced by (ℓ - a) and x is replaced by (ℓ - x). The signs of θ_1 and V_1 are also changed.

**If x is greater than c, x is replaced by (ℓ - x) and c is replaced by (ℓ - c). The signs of y_3 and M_{x3} are also changed.

Remarks:

Deflections must not significantly alter the geometry of the problem. Beams must be of constant cross section for deflection and slope equations to be valid. Stresses must be in the elastic region.

Example 1:

For the beam below, what are the values of deflection, slope, moment, and shear at an x of 114 inches?



Keystrokes:	Display:
[USER]	(set USER mode)
[XEQ] [ALPHA] SIZE [ALPHA] 011	
[XEQ] [ALPHA] FIX [ALPHA]	L?
140 [R/S]	E?

Keystrokes:	Display:
30 [EEX] 6 [R/S]	I?
4.74 [R/S]	a?
0 [R/S]	P?
0 [R/S]	W?
14 [R/S]	M?
147000 [R/S]	c?
30 [R/S]	X?
114 [R/S]	
[A]	Y=43.72E-3
[B]	∠=-3.155E-3
[C]	M=13.05E3
[D]	V=444.7E0

User Instructions

SIZE: 011

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program (USER mode)		[USER]	
2	Begin execution and input data		[XEQ] FIX	L?
		ℓ	[R/S]	E?
		E	[R/S]	I?
		I	[R/S]	a?
		a	[R/S]	P?
		P	[R/S]	W?
		W	[R/S]	M?
		M	[R/S]	C?
		C	[R/S]	X?
		X	[R/S]	
3	Execute desired routine(s)			
	deflection (Y)		[A]	Y=
	slope (θ)		[B]	∠=
	moment (M)		[C]	M=
	shear (V)		[D]	V=
4	For a new point on the beam		[E]	X?
		X	[R/S]	
	Go to step 3			

Program Listings

```
01◆LBL "FIX
"
02 ENG 3
03 "L?"
04 PROMPT
05 STO 02
06 "E?"
07 PROMPT
08 "I?"
09 PROMPT
10 *
11 STO 09
12 "a?"
13 PROMPT
14 STO 06
15 "P?"
16 PROMPT
17 STO 03
18 "W?"
19 PROMPT
20 STO 04
21 "M?"
22 PROMPT
23 STO 05
24 "c?"
25 PROMPT
26 STO 07
27◆LBL E
28 "X?"
29 PROMPT
30 STO 00
31 STOP
32◆LBL B
33 RCL 00
34 SF 00
35 ENTER↑
36 +
37 RCL 02
38 3
39 *
40 RCL 00
41 ENTER↑
42 +
43 GTO 00
44◆LBL A
45 RCL 00
46 CF 00
47 X↑2
48 RCL 02
49 ENTER↑
50 +
```

Initialization

---

angle

deflection

```
51 RCL 00
52◆LBL 00
53 SF 03
54 SF 01
55 -
56 RCL 00
57 *
58 RCL 02
59 X↑2
60 -
61 *
62 24
63 /
64 RCL 04
65 *
66 RCL 06
67 XEQ 01
68 RCL 01
69 *
70 RCL 08
71 RCL 02
72 *
73 3
74 X<>Y
75 *
76 FS? 00
77 LASTX
78 FS? 00
79 -
80 -
81 XEQ 06
82 6
83 /
84 FS? 00
85 3
86 FS? 00
87 *
88 FS? 00
89 XEQ 03
90 XEQ 04
91 RCL 07
92 XEQ 01
93 FC? 00
94 GTO 09
95 3
96 *
97 X<>Y
98 ENTER↑
99 +
100◆LBL 09
101 +
```

angle or deflection

Program Listings

```
102 XEQ 06
103 XEQ 08
104 RCL 09
105 /
106 "Y="
107 FS? 00
108 "∠="
109 GTO d
110◆LBL D
111  SF 00
112  RCL 02
113  RCL 00
114  ENTER↑
115  +
116  GTO 00
117◆LBL C
118  CF 00
119  RCL 02
120  RCL 00
121  -
122  RCL 00
123  *
124  RCL 02
125  X↑2
126  6
127  /
128◆LBL 00
129  SF 03
130  SF 01
131  -
132  2
133  /
134  RCL 04
135  *
136  RCL 06
137  XEQ 01
138  FS? 00
139  GTO 00
140  RCL 01
141  *
142  RCL 08
143  RCL 02
144  *
145  -
146◆LBL 00
147  *
148  FS? 00
149  XEQ 03
150  XEQ 04
151  RCL 07
152  XEQ 01
```

shear

Moment

shear or moment

```
153 FS? 00
154 RDN
155 FS? 00
156 CLX
157 FS? 00
158 RCL 08
159 6
160 *
161 X<>Y
162 ENTER↑
163 +
164 +
165 *
166 CF 03
167◆LBL 08
168 FC? 00
169 XEQ 03
170◆LBL 04
171 RCL 10
172 +
173 FS? 03
174 RTN
175 "M="
176 FS? 00
177 "V="
178 GTO d
179◆LBL 01
180 CF 02
181 STO 08
182 RDN
183 STO 10
184 RCL 08
185 RCL 00
186 STO 01
187 X<=Y?
188 GTO 00
189 SF 02
190 RCL 02
191 RCL 08
192 -
193 STO 08
194 RCL 02
195 RCL 00
196 -
197 STO 01
198◆LBL 00
199 RCL 05
200 FS? 01
201 RCL 03
202 RCL 02
203 RCL 08
204 -
```

store a or c + sum

x beyond loading point?

$\frac{P(\ell-a)^2}{p^3}$ or $\frac{M(\ell-a)}{\ell^3}$

Program Listings

```
205 FS? 01
206 X↑2
207 *
208 RCL 02
209 3
210 Y↑X
211 /
212 RCL 02
213 RCL 08
214 FS? 01
215 GTO 00
216 3
217 *
218 -
219 RCL 02
220 *
221 2
222 /
223 RCL 08
224 RCL 01
225 *
226 RTN
227◆LBL 00
228 ENTER↑
229 +
230 +
231 CF 01
232 RTN
233◆LBL 03
234 FS? 02
235 CHS
236 RTN
237◆LBL 06
238 *
239 RCL 01
240◆LBL 05
241 FC? 00
242 X↑2
243 *
244 RTN
245◆LBL d
246 ARCL X
247 AVIEW
248 STOP
249 .END.
```

$\frac{(\ell-3a)\ \ell}{2}$

(1 + 2a)

sign change

common calculation subroutine

display

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	x	50	
	x, (ℓ-x)		
	ℓ		
	P		
	W		
05	M	55	
	a		
	c		
	a, (ℓ-a); c,(ℓ-c)		
	EI		
10	SUM	60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 011 TOT. REG. 62 USER MODE

ENG 3 FIX ___ SCI ___ ON X OFF ___

DEG ___ RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
00		angle, shear	deflection, moment
01		P	M
02		CHS	continue
03		used	used

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

BEAMS FIXED AT BOTH ENDS

PROGRAM REGISTERS NEEDED: 52

HEWLETT PACKARD
SOLUTION BOOK:
CIVIL ENGINEERING

ROW 1 (1 - 4)

ROW 2 (5 - 12)

ROW 3 (12 - 20)

ROW 4 (21 - 28)

ROW 5 (28 - 37)

ROW 6 (38 - 47)

ROW 7 (48 - 58)

ROW 8 (59 - 68)

ROW 9 (69 - 79)

ROW 10 (80 - 88)

ROW 11 (88 - 93)

ROW 12 (94 - 103)

ROW 13 (103 - 109)

ROW 14 (109 - 117)

ROW 15 (117 - 128)

ROW 16 (129 - 137)

ROW 17 (138 - 148)

ROW 18 (148 - 153)

ROW 19 (154 - 164)

ROW 20 (165 - 173)

ROW 21 (173 - 178)

ROW 22 (179 - 189)

ROW 23 (189 - 200)

ROW 24 (201 - 212)

ROW 25 (213 - 223)

ROW 26 (224 - 234)

ROW 27 (235 - 245)

ROW 28 (246 - 249)

SIMPLY SUPPORTED BEAMS

This program calculates deflection, slope, moment and shear at any specified point along a simply supported beam of uniform cross section. Distributed loads, point loads, applied moments or combinations of all three may be modeled. By using the principle of superposition, complicated beams with multiple point loads, and multiple applied moments can be analyzed.

Equations:



$y = y_1 + y_2 + y_3$ (total deflection)

$y_1 = \frac{P(\ell - a)x}{6EI} [x^2 + (\ell - a)^2 - \ell^2]$* (deflection due to point load)

$y_2 = \frac{-Wx}{24EI} [\ell^3 + x^2 (x - 2\ell)]$ (deflection due to distributed load)

$y_3 = \frac{-Mx}{EI} [c - \frac{x^2}{6\ell} - \frac{\ell}{3} - \frac{c^2}{2\ell}]$** (deflection due to applied moment)

$\theta = \theta_1 + \theta_2 + \theta_3$ (total slope)

$\theta_1 = \frac{P(\ell - a)}{6EI} [3x^2 + (\ell - a)^2 - \ell^2]$* (slope due to point load)

$\theta_2 = -\frac{W}{24EI} [\ell^3 + x^2 (4x - 6\ell)]$ (slope due to distributed load)

$\theta_3 = \frac{-M}{EI} [c - \frac{x^2}{2\ell} - \frac{\ell}{3} - \frac{c^2}{2\ell}]$ ** (slope due to applied moment)

$$M_x = M_{x1} + M_{x2} + M_{x3} \qquad \text{(total moment)}$$

$$M_{x1} = \frac{P(\ell - a)x^*}{\ell} \qquad \text{(moment due to point load)}$$

$$M_{x2} = -\frac{Wx}{2}[x - \ell] \qquad \text{(moment due to distributed load)}$$

$$M_{x3} = \frac{Mx^{**}}{\ell} \qquad \text{(moment due to applied moment)}$$

$$V = V_1 + V_2 + V_3 \qquad \text{(total shear)}$$

$$V_1 = \frac{P(\ell - a)^*}{\ell} \qquad \text{(shear due to point load)}$$

$$V_2 = W\left(\frac{\ell}{2} - x\right) \qquad \text{(shear due to distributed load)}$$

$$V_3 = \frac{M}{\ell} \qquad \text{(shear due to applied moment)}$$

where:

y is the deflection at a distance x from the left support;

θ is the slope (change in y per change in x) at x;

M_x is the moment at x;

V is the shear at x;

I is the moment of inertia of the beam;

E is the modulus of elasticity of the beam;

ℓ is the length of the beam;

P is a concentrated load;

W is a uniformly distributed load with dimensions of force per unit length;

M is an applied moment;

a is the distance from the left support to the point load;

c is the distance to the applied moment.

*If x is greater than a, $(\ell - a)$ is replaced by $-a$ and x is replaced by $(x - \ell)$.

**If x is greater than c, x is replaced by $(x - \ell)$ and c is replaced by $(\ell - c)$.

Remarks:

Deflections must not significantly alter the geometry of the problem. Beams must be of constant cross section for deflection and slope equations to be valid. Stresses must be in the elastic region.

Example 1:

Find the deflection, slope, internal moment and shear at distances of 0, and 60 inches for the beam below. Neglect the weight of the beam.



Keystrokes:	Display:
[USER]	(set USER mode)
[XEQ] [ALPHA] SIZE [ALPHA] 011	
[XEQ] [ALPHA] SIM [ALPHA]	L?
72 [R/S]	E?
30 [EEX] 6 [R/S]	I?
.92 [R/S]	a?
0 [R/S]	P?
0 [R/S]	W?
0 [R/S]	M?
10000 [R/S]	c?
40 [R/S]	X?
0 [R/S]	
[A]	Y=0.000E0
[B]	∠=-1.771E-3
[C]	M=0.000E0
[D]	V=138.9E0
[E]	X?
60 [R/S]	
[A]	Y=2.415E-3
[B]	∠=40.26E-6
[C]	M=-1.667E3
[D]	V=138.9E0

User Instructions

SIZE: 011

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program (USER mode)		[USER]	
2	Begin execution and input data		[XEQ] SIM	L?
		ℓ	[R/S]	E?
		E	[R/S]	I?
		I	[R/S]	a?
		a	[R/S]	P?
		P	[R/S]	W?
		W	[R/S]	M?
		M	[R/S]	c?
		c	[R/S]	X?
		x	[R/S]	
3	Execute desired routine(s)			
	deflection (Y)		[A]	Y=
	slope (θ)		[B]	$\underline{\Delta}$ =
	moment (M)		[C]	M=
	shear (V)		[D]	V=
4	For a new point on the beam		[E]	X?
		x	[R/S]	
	Go to step 3			

Program Listings

```
01◆LBL "SIM
"
02 ENG 3
03 "L?"
04 PROMPT
05 STO 02
06 "E?"
07 PROMPT
08 "I?"
09 PROMPT
10 *
11 STO 09
12 "a?"
13 PROMPT
14 STO 06
15 "P?"
16 PROMPT
17 STO 03
18 "W?"
19 PROMPT
20 STO 04
21 "M?"
22 PROMPT
23 STO 05
24 "c?"
25 PROMPT
26 STO 07
27◆LBL E
28 "X?"
29 PROMPT
30 STO 00
31 STOP
32◆LBL B
33 SF 00
34 GTO 00
35◆LBL A
36 CF 00
37◆LBL 00
38 RCL 02
39 ENTER↑
40 *
41 LASTX
42 *
43 RCL 00
44 FS? 00
45 4
46 FS? 00
47 *
48 RCL 02
49 2
50 *
```

Initialization

Calculate θ or Y

```
51 FS? 00
52 3
53 FS? 00
54 *
55 -
56 RCL 00
57 X↑2
58 *
59 +
60 RCL 04
61 *
62 24
63 /
64 RCL 00
65 X<>Y
66 *
67 FS? 00
68 LASTX
69 CHS
70 XEQ 01
71 RCL 01
72 X↑2
73 FS? 00
74 3
75 FS? 00
76 *
77 RCL 08
78 X↑2
79 +
80 RCL 02
81 X↑2
82 -
83 *
84 6
85 /
86 XEQ 02
87 RCL 01
88 X↑2
89 RCL 02
90 /
91 6
92 /
93 FS? 00
94 3
95 FS? 00
96 *
97 RCL 02
98 3
99 /
100 +
101 RCL 08
```

Program Listings

```
102 XT2
103 2
104 /
105 RCL 02
106 /
107 +
108 RCL 08
109 -
110 *
111 RCL 02
112 *
113 RCL 10
114 +
115 RCL 09
116 /
117 "Y="
118 FS? 00
119 "∠="
120 GTO 05
121◆LBL D
122 SF 00
123 GTO 00
124◆LBL C
125 RCL 00
126 CF 00
127◆LBL 00
128 2
129 /
130 RCL 02
131 FS? 00
132 2
133 FS? 00
134 /
135 RCL 00
136 -
137 RCL 04
138 *
139 *
140 FS? 00
141 LASTX
142 XEQ 01
143 XEQ 02
144 RCL 10
145 +
146 "M="
147 FS? 00
148 "V="
149 GTO 05
150◆LBL 01
151 STO 10
152 RCL 02
```

Calculate V or M

store ℓ-a and x

```
153 RCL 06
154 -
155 STO 08
156 RCL 00
157 STO 01
158 RCL 06
159 X>Y?
160 GTO 00
161 RCL 06
162 CHS
163 STO 08
164 RCL 02
165 ST- 01
166◆LBL 00
167 RCL 03
168 RCL 08
169 *
170 RCL 02
171 /
172 FS? 00
173 RTN
174 RCL 01
175 *
176 RTN
177◆LBL 02
178 RCL 10
179 +
180 STO 10
181 RCL 00
182 STO 01
183 RCL 07
184 STO 08
185 X>Y?
186 GTO 00
187 RCL 00
188 RCL 02
189 -
190 STO 01
191 RCL 02
192 RCL 07
193 -
194 STO 08
195◆LBL 00
196 RCL 05
197 RCL 02
198 /
199 FS? 00
200 RTN
201 RCL 01
202 *
203 RTN
```

$\frac{P(\ell-a)\ x}{\ell}$

store x and c

$\frac{m}{\ell}$

Program Listings

```
204◆LBL 05
205 ARCL X
206 AVIEW
207 STOP
208 .END.
```

Display

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	x	50	
	x, (x − ℓ)		
	ℓ		
	P		
	W		
05	M	55	
	a		
	c		
	(ℓ − a), −a; c, ℓ	− c	
	EI		
10	SUM	60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 011 TOT. REG. 51 USER MODE
ENG 3 FIX ___ SCI ___ ON X OFF ___
DEG ___ RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
00		calculate Y or V	calculate θ or M

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

SIMPLY SUPPORTED BEAMS

PROGRAM REGISTERS NEEDED: 41

HEWLETT PACKARD
SOLUTIONS BOOK:
CIVIL ENGINEERING

ROW 1 (1 : 4)

ROW 2 (5 : 12)

ROW 3 (12 : 20)

ROW 4 (21 : 28)

ROW 5 (28 : 35)

ROW 6 (36 : 46)

ROW 7 (46 : 56)

ROW 8 (57 : 67)

ROW 9 (68 : 76)

ROW 10 (77 : 87)

ROW 11 (88 : 98)

ROW 12 (99 : 111)

ROW 13 (112 : 119)

ROW 14 (120 : 126)

ROW 15 (127 : 137)

ROW 16 (138 : 145)

ROW 17 (146 : 152)

ROW 18 (153 : 164)


ROW 19 (165 : 175)
ROW 20 (176 : 187)
ROW 21 (188 : 199)
ROW 22 (200 : 208)

CANTILEVER BEAMS

This program calculates deflection, slope, moment and shear at any specified point along a rigidly fixed, cantilever beam of uniform cross section. Distributed loads, point loads, applied moments or combinations of all three may be modeled. By using the principle of superposition, complicated beams with multiple point loads, applied moments and combined distributed loads may be analyzed.

Equations:



$$y = y_1 + y_2 + y_3 \qquad \text{(total deflection)}$$

$$y_1 = \frac{PX_1^2}{6EI}(X_1 - 3a) - \frac{Pa^2}{2EI}(x - a)(x > a)^* \qquad \text{(deflection due to point load)}$$

$$y_2 = \frac{-WX_2^2}{6EI}\left[X_2\left(\frac{X_2}{4} - b\right) + 1.5\,b^2\right]$$

$$- \frac{Wb^3}{6EI}(x - b)\ (x > b) \qquad \text{(deflection due to distributed load)}$$

$$y_3 = \frac{MX_3^2}{2EI} + \frac{Mc}{EI}(x - c)\ (x > c) \qquad \text{(deflection due to applied moment)}$$

$$\theta = \theta_1 + \theta_2 + \theta_3 \qquad \text{(total slope)}$$

$$\theta_1 = \frac{PX_1}{2EI}(X_1 - 2a) \qquad \text{(slope due to point load)}$$

$$\theta_2 = \frac{WX_2}{EI}\left[X_2\left(\frac{X_2}{6} - \frac{b}{2}\right) + \frac{b^2}{2}\right] \qquad \text{(slope due to distributed load)}$$

$$\theta_3 = \frac{MX_3}{EI} \qquad \text{(slope due to applied moment)}$$

$$M_x = M_{x1} + M_{x2} + M_{x3} \qquad \text{(total moment)}$$

$$M_{x1} = P(X_1 - a) \qquad \text{(moment due to point load)}$$

$$M_{x2} = -W\,(X_2\,(X_2/2 - b) + b^2/2) \qquad \text{(moment due to distributed load)}$$

$$M_{x3} = M\ (x \leq c) \qquad \text{(moment due to applied moment)}$$

$$V = V_1 + V_2 + V_3 \qquad \text{(total shear)}$$

$$V_1 = P\ (x \leq a) \qquad \text{(shear due to point load)}$$

$$V_2 = W\ (b - X_2) \qquad \text{(shear due to distributed load)}$$

$$V_3 = 0 \qquad \text{(shear due to applied moment)}$$

where:

y is the deflection at a distance x from the wall;
θ is the slope (change in y per change in x) at x;
M_x is the moment at x;
V is the shear at x;
I is the moment of inertia of the beam;
E is the modulus of elasticity of the beam;

ℓ is the length of the beam;

P is a concentrated load;

W is a uniformly distributed load with dimensions of force per unit length

M is an applied moment;

a is the distance from the foundation to the point load;

b is the distance to the end of the distributed load;

c is the distance to the applied moment;

X_1 = x if x $\leq$ a or a if x > a;

X_2 = x if x $\leq$ b or b if x > b;

X_3 = x if x $\leq$ c or c if x > c.

*The notation (x > a) is interpreted as 1.00 if x is greater than a and as 0.00 if x is less than or equal to a.

Remarks:

Deflections must not significantly alter the geometry of the problem. Beams must be of constant cross section for deflection and slope equations to be valid. Stresses must be in the elastic region.

Example 1:

What is the deflection at x = 12? Neglect the weight of the beam.



Keystrokes:	Display:
[USER	(set USER mode)
[XEQ] [ALPHA] SIZE [ALPHA] 011	
[XEQ] [ALPHA] CANT [ALPHA]	L?

Keystrokes:	Display:
15 [R/S]	E?
30 [EEX] 6 [R/S]	I?
4.7 [R/S]	a?
8 [R/S]	P?
100 [R/S]	b?
0 [R/S]	W?
0 [R/S]	c?
0 [R/S]	M?
0 [R/S]	X?
12 [R/S]	
[A]	Y=-211.8E-6

User Instructions

SIZE: 011

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program (USER mode)		[USER]	
2	Begin execution and input data		[XEQ] CANT	L?
		ℓ	[R/S]	E?
		E	[R/S]	I?
		I	[R/S]	a?
		a	[R/S]	P?
		P	[R/S]	b?
		b	[R/S]	W?
		W	[R/S]	c?
		c	[R/S]	M?
		M	[R/S]	X?
		x	[R/S]	
3	Execute desired routine(s)			
	deflection (Y)		[A]	Y=
	slope (θ)		[B]	∠ =
	moment (M)		[C]	M=
	shear (V)		[D]	V=
4	For a new point on the beam		[E]	X?
		x	[R/S]	
	Go to step 3			

Program Listings

```
 01◆LBL "CAN
T"
 02 ENG 3
 03 "L?"
 04 PROMPT
 05 STO 02
 06 "E?"
 07 PROMPT
 08 "I?"
 09 PROMPT
 10 *
 11 STO 10
 12 "a?"
 13 PROMPT
 14 STO 06
 15 "P?"
 16 PROMPT
 17 STO 03
 18 "b?"
 19 PROMPT
 20 STO 07
 21 "W?"
 22 PROMPT
 23 STO 04
 24 "c?"
 25 PROMPT
 26 STO 08
 27 "M?"
 28 PROMPT
 29 STO 05
 30◆LBL E
 31 "X?"
 32 PROMPT
 33 STO 00
 34 STOP
 35◆LBL A
 36 RCL 06
 37 XEQ 04
 38 LASTX
 39 *
 40 CHS
 41 3
 42 *
 43 FS? 02
 44 0
 45 RCL 01
 46 RCL 06
 47 3
 48 *
 49 -
 50 RCL 01
```

Initialization

deflection

```
 51 *
 52 +
 53 RCL 03
 54 *
 55 RCL 01
 56 *
 57 RCL 07
 58 XEQ 04
 59 RCL 07
 60 3
 61 Y↑X
 62 *
 63 FS? 02
 64 CLX
 65 STO 09
 66 RDN
 67 RCL 01
 68 4
 69 /
 70 RCL 07
 71 -
 72 RCL 01
 73 *
 74 RCL 07
 75 X↑2
 76 1.5
 77 *
 78 +
 79 RCL 01
 80 X↑2
 81 *
 82 RCL 09
 83 +
 84 RCL 04
 85 *
 86 -
 87 RCL 08
 88 XEQ 04
 89 6
 90 *
 91 RCL 01
 92 3
 93 *
 94 X<>Y
 95 FS? 02
 96 CLX
 97 +
 98 RCL 05
 99 *
100 RCL 01
101 *
```

Program Listings

```
102 +
103 6
104 /
105 RCL 10
106 /
107 "Y="
108 GTO d
109◆LBL B
110 RCL 06
111 XEQ 04
112 RCL 06
113 2
114 /
115 RCL 01
116 -
117 RCL 03
118 *
119 RCL 01
120 *
121 RCL 07
122 XEQ 04
123 RDN
124 RCL 01
125 6
126 /
127 RCL 07
128 2
129 /
130 -
131 RCL 01
132 *
133 RCL 07
134 X↑2
135 2
136 /
137 +
138 RCL 04
139 *
140 RCL 01
141 *
142 -
143 RCL 08
144 XEQ 04
145 RDN
146 RCL 05
147 RCL 01
148 *
149 +
150 RCL 10
151 /
152 "∠="
```

slope

```
153 GTO d
154◆LBL C
155 RCL 06
156 XEQ 04
157 RCL 01
158 RCL 06
159 -
160 RCL 03
161 *
162 RCL 07
163 XEQ 04
164 CLX
165 RCL 01
166 2
167 /
168 RCL 07
169 -
170 RCL 01
171 *
172 RCL 07
173 X↑2
174 2
175 /
176 +
177 RCL 04
178 *
179 -
180 RCL 08
181 XEQ 04
182 CLX
183 RCL 05
184 X<>Y
185 FS? 02
186 +
187 "M="
188 GTO d
189◆LBL D
190 RCL 06
191 XEQ 04
192 0
193 FS? 02
194 RCL 03
195 RCL 07
196 XEQ 04
197 CLX
198 RCL 01
199 RCL 07
200 -
201 RCL 04
202 *
203 -
```

Moment

shear

Program Listings

```
204 "Y="
205 GTO d
206◆LBL 04
207 CF 02
208 RCL 00
209 STO 01
210 X<>Y
211 X<=Y?
212 STO 01
213 X>Y?
214 SF 02
215 -
216 RTN
217◆LBL d
218 ARCL X
219 AVIEW
220 STOP
221 .END.
```

Display

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	x	50	
	x''(a)		
	ℓ		
	P		
	W		
05	M	55	
	a		
	b		
	c		
	temp storage		
10	FI	60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 011 TOT. REG. 54 USER MODE
ENG 3 FIX ___ SCI ___ ON X OFF ___
DEG ___ RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
02		x > a	x < a

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

CANTILEVER BEAMS

PROGRAM REGISTERS NEEDED: 44



CANTILEVER BEAMS

HEWLETT PACKARD
SOLUTION BOOK:
CIVIL ENGINEERING



BOLT TORQUE

This program may be used to calculate either the torque that will yield a specified bolt load or the load resulting from a specified torque. The maximum shear stress in the body of the screw may also be calculated.

Equations:

$$T = W \frac{D_m}{2} \left[\frac{\tan \alpha + f_t/\cos \theta}{1 - f_t \tan \alpha/\cos \theta} \right] + W f_c \frac{D_c}{2}$$

$$\tau_{max} = \sqrt{(W/2A_r)^2 + (16T_t/\pi D_r^3)^2}$$

$$T_t = T - W f_c \frac{D_c}{2}$$

where:

- T is the applied torque;
- W is the bolt load;
- D_m is the mean thread diameter;
- α is the helix angle of the thread;
- f_t is the coefficient of thread friction;
- θ is one-half of the thread angle;
- f_c is the collar coefficient of friction;
- D_c is the collar diameter;
- τ_{max} is the maximum shear stress in the body of the screw;
- A_r is the root area;
- D_r is the diameter at the root of the thread.

Note:

The accuracy with which f_t and f_c are approximated has a significant effect on the applicability of the resulting computations.

References:

Hall, Holowenko, Laughlin Machine Design, Schaum's Outline Series, McGraw-Hill Co., 1961.

Example:

Some bolts must exert a force of 11,000 pounds each. What torque is necessary to achieve this load assuming the following specifications? What is the shear stress in the bolt?

D_m = 0.3344 in
α = 3.40°
f_t = 0.15
θ = 30°
f_c = 0.30
D_c = 0.8750
D_r = 0.2983

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 010	
[XEQ] [ALPHA] BOLT [ALPHA]	a?
3.4 [R/S]	∠?
30 [R/S]	FT?
0.15 [R/S]	DM?
0.3344 [R/S]	DC?
0.8750 [R/S]	FC?
0.3 [R/S]	W?
11,000 [R/S]	T?
[R/S]	T=1876.03
[XEQ] SHEAR	DR?
0.2983 [R/S]	TMAX=114,335.98

User Instructions

SIZE: 010

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program			
2	Begin execution and input proper values		[XEQ] BOLT	a?
	(skip over unknown with [R/S])	α	[R/S]	Δ ?
		θ	[R/S]	FT?
		f_t	[R/S]	DM?
		D_m	[R/S]	DC?
		D_c	[R/S]	FC?
		f_c	[R/S]	W?
		W	[R/S]	T?
		T	[R/S]	
				T=
				or
				W=
3	To find maximum shear stress		[XEQ] SHEAR	DR?
		D_r	[R/S]	
				TMAX=

Program Listings

```
01◆LBL "BOL
T"
02 FIX 2
03 "a?"
04 PROMPT
05 TAN
06 STO 03
07 "∠?"
08 PROMPT
09 COS
10 STO 02
11 "FT?"
12 PROMPT
13 STO 01
14 RCL 02
15 /
16 RCL 03
17 +
18 1
19 RCL 01
20 RCL 03
21 *
22 RCL 02
23 /
24 -
25 /
26 STO 01
27 "DM?"
28 PROMPT
29 2
30 /
31 STO 06
32 "DC?"
33 PROMPT
34 2
35 /
36 STO 05
37 "FC?"
38 PROMPT
39 STO 04
40 "W?"
41 PROMPT
42 STO 07
43 CLX
44 "T?"
45 PROMPT
46 X=0?
47 GTO B
48 STO 08
49◆LBL A
50 RCL 08
```

Initialization

Common to both W + T

Calculate W

```
51 RCL 01
52 RCL 06
53 *
54 RCL 04
55 RCL 05
56 *
57 +
58 /
59 STO 07
60 "W="
61 GTO 05
62◆LBL B
63 RCL 01
64 RCL 06
65 *
66 RCL 04
67 RCL 05
68 *
69 +
70 RCL 07
71 *
72 STO 08
73 "T="
74 GTO 05
75◆LBL "SHE
AR"
76 "DR?"
77 PROMPT
78 2
79 PI
80 /
81 X<>Y
82 /
83 LASTX
84 /
85 8
86 LASTX
87 /
88 RCL 01
89 RCL 06
90 *
91 RCL 07
92 *
93 *
94 RCL 07
95 R-P
96 X<>Y
97 RDN
98 *
99 "TMAX="
100◆LBL 05
```

Calculate T

Calculate maximum shear stress

Display

Program Listings

```
101 ARCL X
102 AVIEW
103 STOP
104 .END.
```

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00		50	
	f		
	cos θ		
	tan α		
	f		
05	$D^c/2$	55	
	$d^c/2$		
	W^m		
	T		
	temp storage		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 010 TOT. REG. 32 USER MODE

ENG ____ FIX 2 SCI ____ ON ____ OFF X

DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
		NONE	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

BOLT TORQUE

PROGRAM REGISTERS NEEDED: 23

HEWLETT PACKARD
SOLUTION BOOK:
CIVIL ENGINEERING